

Teoria dell'entrata: barriere e deterrenza strategica

Gianmaria Martini

Introduzione

La teoria dell'entrata vuole spiegare perchè nuove imprese non entrano in un mercato in cui le imprese già esistenti (definite come **incumbents**) realizzano profitti superiori ai livelli normali (il mercato offre pertanto tassi di rendimento sul capitale investito maggiori del costo opportunità relativo all'investimento).

Fornisce due spiegazioni, in parte alternative ed in parte complementari, a questo fatto: la teoria delle **barriere all'entrata** e la teoria della **deterrenza strategica all'entrata**.

Le barriere all'entrata, viste in dettaglio durante il corso di Economia Industriale I, possono essere riassunte in una **lista di condizioni di mercato** che spiegano perchè l'entrata non avviene. Tali condizioni possono essere **tecnologiche** (ad es. economie di scala o sunk costs), **legali** (brevetti, diritti di esclusività, tasse governative per l'esercizio di professioni), di **organizzazione del mercato** (canali distributivi, reti di marketing, fedeltà dei consumatori).

La deterrenza strategica all'entrata è invece identificabile in una serie di **azioni** adottate dall' (o dagli) incumbent sottoposto ad una concorrenza potenziale **finalizzate a ridurre** (o eliminare completamente) il rischio che l'entrata si verifichi.

Dal punto di vista del benessere collettivo, la deterrenza strategica è più grave dell'esistenza di barriere all'entrata. Infatti l'aumento del numero delle imprese e quindi della competizione in un mercato è limitato in modo attivo proprio grazie al comportamento dell'incumbent, mentre in presenza di barriere all'entrata non si riscontrano comportamenti finalizzati a impedire l'entrata (anche se le barriere possono essere dovute a precedenti comportamenti strategici).

Modelli di deterrenza all'entrata

I modelli presentati sono i seguenti:

- prezzo limite basato sul Postulato di Sylos Labini
- vincoli di capacità produttiva per impedire l'entrata (Spence)
- rimozione del Postulato di Sylos Labini e scelta della capacità produttiva per prevenire l'entrata (Dixit)
- utilizzo del prezzo limite per segnalare l'efficienza dell'incumbent in presenza di informazione asimmetrica (Milgrom e Roberts)

Prezzo limite nel modello di Bain-Sylos Labini

La maggior parte dei modelli di deterrenza viene formulata su due periodi: un periodo $t = 1$ in cui nel mercato è attivo solo l'incumbent, ed un periodo $t = 2$ in cui un potenziale entrante sceglie se entrare oppure rimanere fuori dal mercato. Nel modello di Sylos Labini vengono formulate le seguenti ipotesi:

- informazione completa
- simmetria nell'efficienza produttiva (incumbent ed entrante posseggono le stesse funzioni di costo - l'analisi può variare introducendo un'asimmetria a favore dell'incumbent)
- Postulato di Sylos Labini: l'entrante assume che l'incumbent continuerà a produrre, anche dopo l'entrata, lo stesso livello di output da lui realizzato a $t = 1$.

Si ipotizzi una funzione di domanda del mercato lineare

$$\begin{aligned} &> \text{p:=a-b*y;} \\ & \qquad \qquad \qquad p := a - b y \end{aligned}$$

ed una funzione di costo dell'incumbent (definita come cti) data da

$$\begin{aligned} &> \text{cti:=c+d*y;} \\ & \qquad \qquad \qquad cti := c + d y \end{aligned}$$

possiamo allora definire l'espressione del profitto dell'incumbent (se rimane monopolista) come la seguente

$$\begin{aligned} &> \text{pii:=p*y-cti;} \\ & \qquad \qquad \qquad pii := (a - b y) y - c - d y \end{aligned}$$

Differenziando tale espressione rispetto a y , risolvendo la condizione del primo ordine rispetto a y otteniamo la quantità di monopolio

```
[ > yi:=solve(diff(pii,y),y);
```

$$yi := \frac{1}{2} \frac{a-d}{b}$$

e quindi sostituendo nell'espressione del profitto, i profitti di monopolio

```
[ > prom:=simplify((a-b*yi)*yi-c-d*yi);
```

$$prom := \frac{1}{4} \frac{a^2 + d^2 - 4 c b - 2 a d}{b}$$

Vediamo ora cosa accade se consideriamo la possibile entrata di un'altra impresa, definita impresa b , la cui funzione di costo è la seguente

```
[ > ctb:=c+d*yb;
```

$$ctb := c + d yb$$

per semplificare l'analisi introduciamo dei dati, che poi potranno essere modificati a piacere, relativi all'intercetta della funzione di domanda del mercato (a), alla pendenza (b), al costo fisso (c) ed al costo medio e marginale (d).

```
[ > data:=[a=1000,b=4,c=10,d=20];
```


$$data := [a = 1000, b = 4, c = 10, d = 20]$$

possiamo rappresentare in un grafico la funzione di costo totale dell'impresa b , sostituendo i dati nella funzione ctb

```
[ > grafctb:=subs(data,ctb);
```

$$grafctb := 10 + 20 yb$$

```
[ > plot(grafctb,yb=0..10, thickness=3);
```



e quindi definire la funzione del costo medio dell'impresa b , ossia cmb

```
> cmb:=(ctb/yb);
```


$$cmb := \frac{c + d yb}{yb}$$

e rappresentarla anch'essa graficamente, dopo aver sostituito per i dati del problema

```
> grafcmb:=subs(data,cmb);
```


$$grafcmb := \frac{10 + 20 yb}{yb}$$

```
> plot(grafcmb,yb=0.2..10, thickness=3);
```



per le ipotesi del modello l'impresa presenta rendimenti crescenti di scala. Le funzioni di costo totale e di costo medio possono essere rappresentate nello stesso grafico

```
> plot([grafctb,grafcmb], yb=0.2..10, thickness=3,
color=[blue,red]);
```



Metodo di soluzione del modello di Sylos Labini

L'adozione del Postulato di Sylos Labini porta il nuovo entrante a ritenere che l'incumbent produrrà comunque la quantità realizzata a $t = 1$ e quindi l'entrante rimarrà con una domanda residuale, data dalla differenza tra la domanda del mercato e la quantità prodotta dall'incumbent.

Sapendo questa cruciale informazione, l'incumbent può allora determinare la posizione della domanda residuale dell'entrante in modo tale che essa porti ad un profitto nullo per l'entrante e quindi ad una non convenienza ad entrare.

Per individuare tale condizione occorre individuare il punto di tangenza tra la curva dei costi medi dell'entrante e la sua domanda residuale.

Per determinare tale punto di tangenza deriviamo la funzione *cmb* per calcolare la sua pendenza.

```
> pendenza:=diff(cmb,yb);
```

$$pendenza := \frac{d}{yb} - \frac{c + d yb}{yb^2}$$

questa pendenza dovrà essere uguale alla pendenza della funzione di domanda residuale, per avere il punto di tangenza.

La pendenza della domanda residuale coincide con quella della domanda del mercato, quindi è pari a $-b$. Uguagliando le due pendenze e risolvendo rispetto a yb otteniamo

```
> soln:=solve(pendenza=-b,yb);
```

$$soln := \frac{\sqrt{c b}}{b}, -\frac{\sqrt{c b}}{b}$$

chiaramente prendiamo la radice positiva, che fornisce il livello di yb in cui costi medi e domanda residuale sono tangenti

```
> coord1:=soln[1];
```

$$coord1 := \frac{\sqrt{c b}}{b}$$

per trovare la domanda residuale, sappiamo che se $y_b = \frac{\sqrt{c b}}{b}$ allora cmb sarà pari al seguente valore

```
> coord2:=d+c/coord1;
```

$$coord2 := d + \frac{c b}{\sqrt{c b}}$$

e per trovare l'intercetta della funzione di domanda residuale risolviamo

```
> interdomresid:=solve(coord2=x-b*coord1, x);
```

$$interdomresid := \frac{d \sqrt{c b} + 2 c b}{\sqrt{c b}}$$

la funzione di domanda residuale è dunque la seguente

```
> domresid:=interdomresid-b*yb;
```

$$domresid := \frac{d \sqrt{c b} + 2 c b}{\sqrt{c b}} - b y b$$

l'espressione del profitto dell'entrante è dunque data da

```
> pib:=(interdomresid-b*ybb)*ybb-c-d*ybb;
```

$$pib := \left(\frac{d \sqrt{c b} + 2 c b}{\sqrt{c b}} - b y b b \right) y b b - c - d y b b$$

e se sostituiamo per l'output da lui prodotto il valore in cui domanda residuale e costi medi sono tangenti otteniamo

```
> ybb:=coord1;
```

$$ybb := \frac{\sqrt{c b}}{b}$$

```
> simplify(pib);
```

0

un profitto nullo, che lo induce dunque a non entrare.

Se risolviamo il nostro caso numerico otteniamo

```
> casonum:=subs(data,domresid);
```

$$casonum := \frac{1}{40} (20 \sqrt{40} + 80) \sqrt{40} - 4 yb$$


il profitto dell'entrante è

```
> probcasonum:=casonum*yb-grafctb;
```

$$probcasonum := \left(\frac{1}{40} (20 \sqrt{40} + 80) \sqrt{40} - 4 yb \right) yb - 10 - 20 yb$$


che possiamo rappresentare graficamente e notare che al massimo esso sarà pari a 0

```
> plot(probcasonum,yb=0..2, thickness=3);
```



possiamo anche rappresentare insieme la domanda residuale e la funzione dei costi medi dell'entrante ed ottenere il punto di tangenza

```
> plot([casonum,grafcmb], yb=0.2..5, thickness=3,
color=[blue,red]);
```

a questo punto occorre determinare la quantità limite (dato che il prezzo limite è l'intercetta della funzione di domanda residuale). Essa si ottiene invertendo la funzione di domanda e sostituendo il prezzo limite per p , ottenendo

```
> ylim:=a/b-(1/b)*interdomresid;
```

$$y_{lim} := \frac{a}{b} - \frac{d\sqrt{cb} + 2cb}{b\sqrt{cb}}$$

L'espressione della domanda residuale nel caso numerico è dunque

```
> domresidfunzylim:=subs(data,a-b*ymm-b*yb);
```

$$domresidfunzylim := 1000 - 4ymm - 4yb$$

e possiamo vedere graficamente qual'è l'effetto della domanda residuale e quindi di diversi livelli di prezzo limite, sul profitto dell'entrante

```
> with(plots):
```

Warning, the name changecoords has been redefined

definiamo un grafico che mostra come cambia la domanda residuale se cambia l'output dell'incumbent

```
> graf1:=animate(domresidfunzylim, yb=0.2..10,
```


```
ymm=200..250,color=blue,thickness=3):
```

definiamo un grafico che rappresenta la funzione del costo medio dell'entrante

```
> graf2:=plot(grafcmb, yb=0.2..10,color=red,thickness=3):
```

possiamo adesso rappresentarli insieme e vedere l'effetto di diversi output limite (e quindi diversi prezzi limite)

```
> display([graf1,graf2]);
```




e' anche possibile vedere l'effetto in 3D sul profitto dell'entrante di livelli crescenti di output dell'incumbent

definiamo il profitto dell'entrante in funzione di *ymm* (output incumbent) e *yb* output entrante

```
> prob:=domresidfunzylim*yb-subs(data,ctb);
```

```
prob := (1000 - 4 ymm - 4 yb) yb - 10 - 20 yb
```

```
> plot3d(prob,yb=0..10,ymm=0..500);
```



si nota come al crescere di y_{mm} il profitto dell'entrante diminuisce.

E' interessante definire il profitto dell'incumbent ottenuto tramite il prezzo limite

```
> prolim:=simplify(interdomresid*y_lim-c-d*y_lim);
```

$$prolim := \frac{-2d\sqrt{cb} + 2a\sqrt{cb} - 5cb}{b}$$

sostituendo alcuni dati si ottiene

```
> data1:=[b=4,c=10,d=20];
```


$$data1 := [b = 4, c = 10, d = 20]$$

e possiamo allora confrontare, al variare dell'intercetta della funzione di domanda, il profitto dell'incumbent in assenza di minaccia di entrata e quello in presenza di minaccia di entrata

```
> grafconfr:=subs(data1, [prom,prolim]);
```

$$grafconfr := \left[\frac{1}{16}a^2 + 15 - \frac{5}{2}a, -10\sqrt{40} + \frac{1}{2}a\sqrt{40} - 50 \right]$$

```
> plot(grafconfr, a=100..1000, color=[blue,red],thickness=3);
```



E' anche possibile osservare cosa accade al variare del costo marginale e medio

```
> data2:=[a=1000,c=20];
```

```
data2 := [a = 1000, c = 20]
```

L'espressione del profitto di monopolio diventa

```
> graf2:=subs(data2,prom);
```

$$\text{graf2} := \frac{1}{4} \frac{1000000 + d^2 - 80 b - 2000 d}{b}$$

e quella del profitto ottenuto per mezzo del prezzo limite

```
> graf3:=subs(data2,prolim);
```

$$\text{graf3} := \frac{-2 d \sqrt{20} \sqrt{b} + 2000 \sqrt{20} \sqrt{b} - 100 b}{b}$$

possiamo adesso costruire dei grafici che mostrano cosa accade alla differenza dei profitti se varia sia il costo marginale di produzione che la pendenza della curva di domanda

```
> graf3:=animate(graf2, b=1..10, d=1..100, color=blue,
thickness=3):
```

```
> graf4:=animate(grafc3, b=1..10, d=1..100,  
color=red,thickness=3):  
> display([graf3,graf4]);
```

